

CRITERI di ARRESTO in $F(\beta, m)$

- $h, [a, b]$ che verif in (1) e (2) teo conv
- $\varphi: F(\beta, m) \rightarrow F(\beta, m)$ t.c:

$$\exists \delta > 0 : \forall \xi \in [a, b] \cap F(\beta, m), |\varphi(\xi) - h(\xi)| \leq \delta$$
- la success generata dal m. it def da φ a partire da ξ_0 e' in $[a, b]$

$$\Rightarrow |\xi_k - \alpha| \leq L^k \left(|\xi_0 - \alpha| - \frac{\delta}{1-L} \right) + \frac{\delta}{1-L}$$

" ξ_k va a finire nell'int di centro α e raggio $\frac{\delta}{1-L}$ "

(1) dato $\epsilon > 0$: $\exists E$ $|\varphi(\xi_k) - \xi_k| < E$ ALLORA STOP

$$\begin{aligned} \bullet \quad \varphi(\xi_k) - \xi_k &= \varphi(\xi_k) - h(\xi_k) + \\ &+ h(\xi_k) - h(\alpha) + \\ &+ \alpha - \xi_k \end{aligned}$$


$$\begin{aligned} &= \varphi(\xi_k) - h(\xi_k) + h'(\theta_k)(\xi_k - \alpha) \\ &\quad - (\xi_k - \alpha) \end{aligned}$$

$$\Rightarrow \xi_k - \alpha = \frac{\varphi(\xi_k) - \xi_k - (\varphi(\xi_k) - h(\xi_k))}{h'(\theta_k) - 1}$$

$$\Rightarrow |\xi_k - \alpha| \leq \frac{|\varphi(\xi_k) - \xi_k| + \delta}{1-L}$$

$$\bullet \quad |\varphi(\xi_k) - \xi_k| < \epsilon \Rightarrow |\xi_k - \alpha| < \frac{\epsilon + \delta}{1-L}$$

$$\bullet \quad \text{se } \epsilon \leq \frac{2\delta}{1-L} \text{ il cr d'arresto pu\`o'}$$

risultare NON efficace ... 

• f con f' continua e $\neq 0$ in $[a, b]$ contenenti α zero di f

• $\psi: F(\beta, m) \rightarrow F(\beta, m)$ t.c.

$$\exists \delta: \forall \xi \in [a, b] \cap F(\beta, m), |f(\xi) - \psi(\xi)| \leq \delta$$

• la success $\xi_k \dots$ \rightarrow in $[a, b]$

(2) dato $\epsilon > 0$, SE $|\psi(\xi_k)| < \epsilon$ ALLORA STOP

$$\begin{aligned} \bullet \quad \psi(\xi_k) &= \psi(\xi_k) - f(\xi_k) + f(\xi_k) - f(\alpha) \\ &= \psi(\xi_k) - f(\xi_k) + f'(\vartheta_k)(\xi_k - \alpha) \end{aligned}$$

$$\Rightarrow \xi_k - \alpha = \frac{\psi(\xi_k) - (\psi(\xi_k) - f(\xi_k))}{f'(\vartheta_k)}$$

$$\Rightarrow |\xi_k - \alpha| \leq \frac{|\Psi(\xi_k)| + \delta}{m} \quad \left(m = \min_{x \in [a, b]} |f'(x)| \right)$$

$$\bullet |\Psi(\xi_k)| < \epsilon \Rightarrow |\xi_k - \alpha| \leq \frac{\epsilon + \delta}{m}$$

$$\bullet \text{ se } \epsilon \leq M_0 + \delta \quad \left(M_0 = \max_{x \in \left\{ |x - \alpha| \leq \frac{\delta}{1-L} \right\}} |f(x)| \right)$$

il cr d'arresto puo' risultare

NON efficace

Es realizzazione in Scilab del m. it
def da:

$$h(x) = e^{-x}$$

```

0001 // Metodi ad un punto
0002 //
0003 // Funzione che definisce il metodo
0004 //
0005 function y=H(x)
0006     y = exp(-x)
0007 endfunction
0008 //
0009 // Valore iniziale, errore richiesto e numero massimo di iterazioni
0010 //
0011 x0 = 1/2;
0012 E = 1d-16;
0013 KMAX = 100;
0014 //
0015 // inizio iterazioni
0016 //
0017 x = x0;
0018 x($+1) = H(x($));
0019 k = 1; // indice della successione
0020 while abs(x($) - x($-1)) > E & k < KMAX,
0021     x($+1) = H(x($));
0022     k = k+1;
0023 end;
0024 //
0025 // fine iterazioni
0026 //
0027 if k == KMAX then info = 2; else info = 1; end;
0028 printf('\n*** info = %d , x(%d) = %15.6e\n', info, k, x($));
0029 // genera nella console le tre righe:
0030 // (1)
0031 // (2)*** info = 1 , x(63) =    5.671433e-01
0032 // (3)
0033 // La 'stringa di formattazione' \n significa 'a capo'. Le stringhe %d e
0034 // %15.6e specificano il formato di visualizzazione del valore delle
0035 // corrispondenti variabili: %d specifica che il valore
0036 // (un numero intero) deve essere scritto utilizzando la usuale notazione
0037 // posizionale in base dieci; %15.6e specifica che il valore (un numero reale)
0038 // deve essere scritto, utilizzando almeno 15 caratteri in totale
0039 // (eventualmente inserendo spazi in testa), nella forma:
0040 //           [segno] c(0).c(1)...c(6) e [segno] esponente
0041 // (il 6 che compare nel formato indica il numero di cifre da utilizzare
0042 // dopo il punto decimale).
0043 //
0044 // grafici
0045 //
0046 clf();
0047 subplot(211);
0048 // La finestra grafica viene suddivisa 2 sottofinestre
0049 // allineate verticalmente:
0050 //
0051 //           +-----+
0052 //           | sottof numero 1 |
0053 //           +-----+
0054 //           | sottof numero 2 |
0055 //           +-----+
0056 //
0057 // e viene selezionata la sottofinestra numero 1
0058 t = linspace(0,1,200)';
0059 plot2d(t,[H(t),t], style=[2,5]);
0060 xgrid(); xlabel('x'); ylabel('y'); legend('y = H(x)', 'y = x');
0061 plot2d(x($),H(x($)), style = -2);
0062 //
0063 subplot(212); // viene selezionata la sottofinestra numero 2
0064 L = exp(-1/2);
0065 d0 = abs(x0 - x($));
0066 indice = (0:k)';
0067 plot2d('nl',indice,L^(indice) * d0, style = 5);
0068 plot2d('nl',indice,abs(x - x($)), style = 2);
0069 xgrid(); xlabel('k'); ylabel('y');
0070 //
0071 legend('L^k * |x(0) - x($)|', '|x(k) - x($)|');

```

